

NAG Library Routine Document

S30FAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

S30FAF computes the price of a standard barrier option.

2 Specification

```

SUBROUTINE S30FAF (CALPUT, TYPE, M, N, X, S, H, K, T, SIGMA, R, Q, P,      &
                  LDP, IFAIL)
INTEGER           M, N, LDP, IFAIL
REAL (KIND=nag_wp) X(M), S, H, K, T(N), SIGMA, R, Q, P(LDP,N)
CHARACTER(1)     CALPUT
CHARACTER(2)     TYPE

```

3 Description

S30FAF computes the price of a standard barrier option, where the exercise, for a given strike price, X , depends on the underlying asset price, S , reaching or crossing a specified barrier level, H . Barrier options of type **In** only become active (are knocked in) if the underlying asset price attains the pre-determined barrier level during the lifetime of the contract. Those of type **Out** start active and are knocked out if the underlying asset price attains the barrier level during the lifetime of the contract. A cash rebate, K , may be paid if the option is inactive at expiration. The option may also be described as **Up** (the underlying price starts below the barrier level) or **Down** (the underlying price starts above the barrier level). This gives the following options which can be specified as put or call contracts.

Down-and-In: the option starts inactive with the underlying asset price above the barrier level. It is knocked in if the underlying price moves down to hit the barrier level before expiration.

Down-and-Out: the option starts active with the underlying asset price above the barrier level. It is knocked out if the underlying price moves down to hit the barrier level before expiration.

Up-and-In: the option starts inactive with the underlying asset price below the barrier level. It is knocked in if the underlying price moves up to hit the barrier level before expiration.

Up-and-Out: the option starts active with the underlying asset price below the barrier level. It is knocked out if the underlying price moves up to hit the barrier level before expiration.

The payoff is $\max(S - X, 0)$ for a call or $\max(X - S, 0)$ for a put, if the option is active at expiration, otherwise it may pay a pre-specified cash rebate, K . Following Haug (2007), the prices of the various standard barrier options can be written as shown below. The volatility, σ , risk-free interest rate, r , and annualised dividend yield, q , are constants. The integer parameters, j and k , take the values ± 1 , depending on the type of barrier.

$$\begin{aligned}
 A &= jSe^{-qT}\Phi(jx_1) - jXe^{-rT}\Phi(j[x_1 - \sigma\sqrt{T}]) \\
 B &= jSe^{-qT}\Phi(jx_2) - jXe^{-rT}\Phi(j[x_2 - \sigma\sqrt{T}]) \\
 C &= jSe^{-qT}\left(\frac{H}{S}\right)^{2(\mu+1)}\Phi(ky_1) - jXe^{-rT}\left(\frac{H}{S}\right)^{2\mu}\Phi(k[y_1 - \sigma\sqrt{T}]) \\
 D &= jSe^{-qT}\left(\frac{H}{S}\right)^{2(\mu+1)}\Phi(ky_2) - jXe^{-rT}\left(\frac{H}{S}\right)^{2\mu}\Phi(k[y_2 - \sigma\sqrt{T}]) \\
 E &= Ke^{-rT}\left\{\Phi(k[x_2 - \sigma\sqrt{T}]) - \left(\frac{H}{S}\right)^{2\mu}\Phi(k[y_2 - \sigma\sqrt{T}])\right\} \\
 F &= K\left\{\left(\frac{H}{S}\right)^{\mu+\lambda}\Phi(kz) + \left(\frac{H}{S}\right)^{\mu-\lambda}\Phi(k[z - \sigma\sqrt{T}])\right\}
 \end{aligned}$$

with

$$\begin{aligned}
x_1 &= \frac{\ln(S/X)}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
x_2 &= \frac{\ln(S/H)}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
y_1 &= \frac{\ln(H^2/(SX))}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
y_2 &= \frac{\ln(H/S)}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
z &= \frac{\ln(H/S)}{\sigma\sqrt{T}} + \lambda\sigma\sqrt{T} \\
\mu &= \frac{r - q - \sigma^2/2}{\sigma^2} \\
\lambda &= \sqrt{\mu^2 + \frac{2r}{\sigma^2}}
\end{aligned}$$

and where Φ denotes the cumulative Normal distribution function,

$$\Phi(x) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^x \exp(-y^2/2) dy.$$

Down-and-In ($S > H$):

When $X \geq H$, with $j = k = 1$,

$$P_{\text{call}} = C + E$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = B - C + D + E$$

When $X < H$, with $j = k = 1$

$$P_{\text{call}} = A - B + D + E$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = A + E.$$

Down-and-Out ($S > H$):

When $X \geq H$, with $j = k = 1$,

$$P_{\text{call}} = A - C + F$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = A - B + C - D + F$$

When $X < H$, with $j = k = 1$,

$$P_{\text{call}} = B - D + F$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = F.$$

Up-and-In ($S < H$):

When $X \geq H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = A + E$$

and with $j = k = -1$,

$$P_{\text{put}} = A - B + D + E$$

When $X < H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = B - C + D + E$$

and with $j = k = -1$,

$$P_{\text{put}} = C + E.$$

Up-and-Out ($S < H$):

When $X \geq H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = F$$

and with $j = k = -1$,

$$P_{\text{put}} = B - D + F$$

When $X < H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = A - B + C - D + F$$

and with $j = k = -1$,

$$P_{\text{put}} = A - C + F.$$

The option price $P_{ij} = P(X = X_i, T = T_j)$ is computed for each strike price in a set X_i , $i = 1, 2, \dots, m$, and for each expiry time in a set T_j , $j = 1, 2, \dots, n$.

4 References

Haug E G (2007) *The Complete Guide to Option Pricing Formulas* (2nd Edition) McGraw-Hill

5 Parameters

1: CALPUT – CHARACTER(1)

Input

On entry: determines whether the option is a call or a put.

CALPUT = 'C'

A call; the holder has a right to buy.

CALPUT = 'P'

A put; the holder has a right to sell.

Constraint: CALPUT = 'C' or 'P'.

2: TYPE – CHARACTER(2)

Input

On entry: indicates the barrier type as **In** or **Out** and its relation to the price of the underlying asset as **Up** or **Down**.

TYPE = 'DI'

Down-and-In.

TYPE = 'DO'

Down-and-Out.

TYPE = 'UI'

Up-and-In.

TYPE = 'UO'

Up-and-Out.

Constraint: TYPE = 'DI', 'DO', 'UI' or 'UO'.

- 3: M – INTEGER *Input*
On entry: the number of strike prices to be used.
Constraint: $M \geq 1$.
- 4: N – INTEGER *Input*
On entry: the number of times to expiry to be used.
Constraint: $N \geq 1$.
- 5: X(M) – REAL (KIND=nag_wp) array *Input*
On entry: X(*i*) must contain X_i , the *i*th strike price, for $i = 1, 2, \dots, M$.
Constraint: $X(i) \geq z$ and $X(i) \leq 1/z$, where $z = X02AMF()$, the safe range parameter, for $i = 1, 2, \dots, M$.
- 6: S – REAL (KIND=nag_wp) *Input*
On entry: S, the price of the underlying asset.
Constraint: $S \geq z$ and $S \leq 1.0/z$, where $z = X02AMF()$, the safe range parameter.
- 7: H – REAL (KIND=nag_wp) *Input*
On entry: the barrier price.
Constraint: $H \geq z$ and $H \leq 1/z$, where $z = X02AMF()$, the safe range parameter.
- 8: K – REAL (KIND=nag_wp) *Input*
On entry: the value of a possible cash rebate to be paid if the option has not been knocked in (or out) before expiration.
Constraint: $K \geq 0.0$.
- 9: T(N) – REAL (KIND=nag_wp) array *Input*
On entry: T(*i*) must contain T_i , the *i*th time, in years, to expiry, for $i = 1, 2, \dots, N$.
Constraint: $T(i) \geq z$, where $z = X02AMF()$, the safe range parameter, for $i = 1, 2, \dots, N$.
- 10: SIGMA – REAL (KIND=nag_wp) *Input*
On entry: σ , the volatility of the underlying asset. Note that a rate of 15% should be entered as 0.15.
Constraint: SIGMA > 0.0.
- 11: R – REAL (KIND=nag_wp) *Input*
On entry: r , the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.
Constraint: $R \geq 0.0$.
- 12: Q – REAL (KIND=nag_wp) *Input*
On entry: q , the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.
Constraint: $Q \geq 0.0$.
- 13: P(LDP,N) – REAL (KIND=nag_wp) array *Output*
On exit: P(*i*,*j*) contains P_{ij} , the option price evaluated for the strike price X_i at expiry T_j for $i = 1, 2, \dots, M$ and $j = 1, 2, \dots, N$.

14: LDP – INTEGER *Input*

On entry: the first dimension of the array P as declared in the (sub)program from which S30FAF is called.

Constraint: $LDP \geq M$.

15: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, CALPUT = $\langle value \rangle$ was an illegal value.

IFAIL = 2

On entry, TYPE = $\langle value \rangle$ was an illegal value.

IFAIL = 3

On entry, M = $\langle value \rangle$.
Constraint: $M \geq 1$.

IFAIL = 4

On entry, N = $\langle value \rangle$.
Constraint: $N \geq 1$.

IFAIL = 5

On entry, X($\langle value \rangle$) = $\langle value \rangle$.
Constraint: $X(i) \geq \langle value \rangle$ and $X(i) \leq \langle value \rangle$.

IFAIL = 6

On entry, S = $\langle value \rangle$.
Constraint: $S \geq \langle value \rangle$ and $S \leq \langle value \rangle$.

IFAIL = 7

On entry, H = $\langle value \rangle$.
Constraint: $H \geq \langle value \rangle$ and $H \leq \langle value \rangle$.

IFAIL = 8

On entry, K = $\langle value \rangle$.
Constraint: $K \geq 0.0$.

IFAIL = 9

On entry, $T(\langle value \rangle) = \langle value \rangle$.
 Constraint: $T(i) \geq \langle value \rangle$.

IFAIL = 10

On entry, $SIGMA = \langle value \rangle$.
 Constraint: $SIGMA > 0.0$.

IFAIL = 11

On entry, $R = \langle value \rangle$.
 Constraint: $R \geq 0.0$.

IFAIL = 12

On entry, $Q = \langle value \rangle$.
 Constraint: $Q \geq 0.0$.

IFAIL = 14

On entry, $LDP = \langle value \rangle$ and $M = \langle value \rangle$.
 Constraint: $LDP \geq M$.

IFAIL = 15

On entry, S and H are inconsistent with TYPE: $S = \langle value \rangle$ and $H = \langle value \rangle$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
 See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
 See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.
 See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The accuracy of the output is dependent on the accuracy of the cumulative Normal distribution function, Φ . This is evaluated using a rational Chebyshev expansion, chosen so that the maximum relative error in the expansion is of the order of the *machine precision* (see S15ABF and S15ADF). An accuracy close to *machine precision* can generally be expected.

8 Parallelism and Performance

S30FAF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example computes the price of a Down-and-In put with a time to expiry of 6 months, a stock price of 100 and a strike price of 100. The barrier value is 95 and there is a cash rebate of 3, payable on expiry if the option has not been knocked in. The risk-free interest rate is 8% per year, there is an annual dividend return of 4% and the volatility is 30% per year.

10.1 Program Text

```

Program s30faf

!      S30FAF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: nag_wp, s30faf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: h, k, q, r, s, sigma
Integer                    :: i, ifail, j, ldp, m, n
Character (1)              :: calput
Character (2)              :: type
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: p(:,,:), t(:,), x(:)
!      .. Executable Statements ..
Write (nout,*) 'S30FAF Example Program Results'

!      Skip heading in data file
Read (nin,*)

Read (nin,*) calput, type
Read (nin,*) s, h, k, sigma, r, q
Read (nin,*) m, n

ldp = m
Allocate (p(ldp,n),t(n),x(m))

Read (nin,*)(x(i),i=1,m)
Read (nin,*)(t(i),i=1,n)

ifail = 0
Call s30faf(calput,type,m,n,x,s,h,k,t,sigma,r,q,p,ldp,ifail)

Write (nout,*)
Write (nout,*) 'Standard Barrier Option'

Select Case (calput)
Case ('C','c')
  Write (nout,*) 'Call :'
Case ('P','p')
  Write (nout,*) 'Put :'
End Select

Select Case (type)
Case ('DI','di','Di','dI')
  Write (nout,*) 'Down-and-In'
Case ('DO','do','Do','dO')
  Write (nout,*) 'Down-and-Out'
Case ('UI','ui','Ui','uI')
  Write (nout,*) 'Up-and-In'

```

```

Case ('UO','uo','Uo','uO')
  Write (nout,*) 'Up-and-Out'
End Select

Write (nout,99998) ' Spot      = ', s
Write (nout,99998) ' Barrier   = ', h
Write (nout,99998) ' Rebate    = ', k
Write (nout,99998) ' Volatility = ', sigma
Write (nout,99998) ' Rate      = ', r
Write (nout,99998) ' Dividend  = ', q

Write (nout,*)
Write (nout,*) ' Strike      Expiry      Option Price'

Do i = 1, m

  Do j = 1, n
    Write (nout,99999) x(i), t(j), p(i,j)
  End Do

End Do

99999 Format (1X,2(F9.4,1X),6X,F9.4)
99998 Format (A,1X,F8.4)
End Program s30fafa

```

10.2 Program Data

S30FAF Example Program Data

'P'	'DI'	:	Call = 'C', Put = 'P', Type				
100.0	95.0	3.0	0.3	0.08	0.04	:	S, H, K, SIGMA, R, Q
1	1	:	M, N				
100.0	:	X(I), I = 1,2,...M					
0.5	:	T(I), I = 1,2,...N					

10.3 Program Results

S30FAF Example Program Results

Standard Barrier Option

Put :

Down-and-In

Spot	=	100.0000
Barrier	=	95.0000
Rebate	=	3.0000
Volatility	=	0.3000
Rate	=	0.0800
Dividend	=	0.0400

Strike	Expiry	Option Price
100.0000	0.5000	7.7988
